

BINGO

1. DESCRIPCION:

En la línea 50 se dimensionan 4 matrices, cada una de las cuales va a tener el siguiente cometido:

* núm (4, 3, 5): Contiene los números de cada cartón (El primer dígito corresponde al número del jugador; el segundo al número de fila dentro de cada cartón; y el tercero, al número de columna de cada cartón). Hay que resaltar el hecho de que cada cartón esta formado por 15 números distribuidos en 3 filas de 5 columnas cada una.

* fila (4,3): Posee la cantidad de número que hay sin tapar en cada fila de cada jugador. Como cada una posee 5 números, inicialmente se da el valor de 5. A medida que van surgiendo números, se van restando unidades a esta matriz; de manera que cuando valga cero, se ha completado una fila (cantándolo).

* tot (4): Juega el mismo papel que la anterior, pero en vez de controlar la fila, lo que hace es informar sobre el posible bingo. Por ello se inicializa a 15 (línea 140).

* número (100): Posee un 1 si el número resultante ha sido extraído. Almacenará un cero cuando dicho número, aún, no se haya cantado.

* cartón (4,100): Declarado en la línea 60, tiene por misión impedir que un número se repita en el mismo cartón. si contiene un 1, es que ya ha salido. En caso contrario, poseerá un cero.

Las líneas 70-110 especifican las ventanas que van a ser usadas: Número 1 para el primer jugador; la 2 para el jugador 2 y así sucesivamente. La número 5 es utilizada para presentar el número que sale u otros mensajes.

De la línea 120 a 170 se inician las tablas que van a controlar el bingo y la línea (La primera a 15 pues son esos los números que hay por carton y la segunda a 5 que coincide con los números existentes en cada línea.

Entre la 200 y la 260 se lleva a cabo la creación de los cartones: Se calcula un número aleatorio entre 1 y 99 imprimiéndose en el cartón correspondiente que indica la variable "i".

El número que se va cantando a lo largo del juego está en la línea 280 (variable "res"). Si este nú-

mero ya ha salido, implica que número (res) = 1 por lo que se repetiría el proceso para calcular otro. Si no hubiera salido, se almacena 1 en la posición correspondiente para que no se repita de nuevo.

Las líneas 340-440 tienen la siguiente misión:

Si cada número de cada cartón que juega el ordenador (num(i, j, k) es distinto del que se ha obtenido (res) y es distinto de cero (implica que ese número aún no se ha tapado) se imprime tal como está (línea 430).

Si coincide el número del cartón con el obtenido se resta al contador de números por cartón y por línea una unidad.

En este último caso, se iguala num (i, j, k) a cero para indicar que ese número ya ha sido tapado y se imprime, en su lugar, un cuadrado oscuro (v\$. Correspondiente al valor 207 del código ASCII).

Una vez actualizadas las tablas de bingo y línea pasa a comprobarse si son cero para cantar lo que corresponda (líneas 400 y 410).

A partir de la línea 460 y hasta la 570 viene reflejado el turno del jugador número 4 (que somos nosotros y corresponde al cartón de más abajo).

PROGRAMA comentado en BASIC

Las posibilidades que este tiene son 3:

* Pulsar "s": El ordenador buscará en nuestro cartón, el número que ha salido. Si lo encuentra lo tapa (línea 520).

* Pulsar "l": En el momento de ir a cantar línea (línea 540).

* Pulsar "b": Cuando vayamos a cantar bingo (línea 530).

La 550 imprime el cuadrado oscuro al encontrar el número.

La 560 imprime el número que no corresponde con el obtenido, tal y como está.

Las líneas 600-620 son la subrutina de "línea". Su objetivo el de informarnos a través de la ventana 1 quien ha sido el afortunado que ha cantado línea e impedir que otro jugador vuelva a cantarla (línea = 1).

De 640-68* es la subrutina de Bingo y lo que se consigue en las líneas 810-final es hacer sonar una música de felicitación.

2. FUNCIONAMIENTO

El juego consiste en simular el juego del bingo tratando de cantar línea y bingo.

El ordenador controla los tres primeros cartones mientras que nosotros hemos de rellenar el cuarto (situado en la parte más baja de la pantalla). Para ello disponemos de tres controles:

* "s" para tapar el número de

nuestro cartón y que coincide con el presentado en la esquina superior izquierda.

* "l" se ha de pulsar cuando aparece (en la esquina superior izquierda) el número que nos queda para rellenar una fila siempre y cuando NADIE haya cantado "línea" hasta entonces. No hace falta pulsar, también, la "s".

* "b" se usa de forma parecida a "l" sólo que para el bingo. Tampoco hace falta pulsar la "s".

Nota: Modificando la línea 320 podemos adaptar la velocidad con que salen los números a nuestro gusto. Si se quiere que vaya más rápido habrá que suprimirla, mientras que si lo que pretendemos es que sea más lento, habrá que poner 200, 300... lo que se quiera en vez de 50 (Bucle "FOR").

```

10 REM *****
20 REM ***** PROGRAMA DE 'BINGO': ANGEL LARUMBE *****
30 REM *****
40 GOSUB 780
50 MODE 2: DIM num(4,3,5), fila(4,3), tot(4), numero(100): v$=CHR$(207)
60 DIM carton(4,100)
70 WINDOW#1,49,70,10,17
80 WINDOW#2,29,50,2,9
90 WINDOW#3,9,30,10,17
100 WINDOW#4,29,50,16,24
110 WINDOW#5,1,80,1,1
120 REM ----- INICIALIZACION DE TABLAS PARA CONTROL DE LINEA Y BINGO -----
130 FOR i=1 TO 4
140 tot(i)=15
150 FOR j=1 TO 3
160 fila(i,j)=5
170 NEXT j,i
180 FOR i=1 TO 4:PAPER#i,3: PEN#i,0: CLS#i:NEXT i
190 REM ----- REPARTO DE CARTONES -----
200 FOR i=1 TO 4
210 PRINT#i," "
220 FOR j=1 TO 3
230 FOR k=1 TO 5
240 num(i,j,k)=INT(RND*99)+1: IF carton(i,num(i,j,k))=1 THEN 240
250 PRINT#i," ";:PRINT#i,USING "###";num(i,j,k);:carton(i,num(i,j,k))=1
260 NEXT k:PRINT#i," ":PRINT#i," ":NEXT j,i
270 REM ----- ELECCION DE UN NUMERO -----
280 res=INT(RND*99)+1:CLS#5
290 IF numero(res)<>1 THEN PRINT#5,"Numero: ";res:SOUND 1,10,15
300 IF numero(res)=1 THEN 280
310 numero(res)=1
320 FOR z=1 TO 50:NEXT
330 REM ----- CONTROL DE LOS CARTONES QUE JUEGA EL ORDENADOR -----

```

PROGRAMA
comentado en
BASIC

```
340 FOR i=1 TO 3
350 CLS#i:PRINT#i," "
360 FOR j=1 TO 3
370 FOR k=1 TO 5
380 IF num(i,j,k)<>res AND num(i,j,k)<>0 THEN 430
390 IF num(i,j,k)=res THEN fila(i,j)=fila(i,j)-1:tot(i)=tot(i)-1
400 IF tot(i)=0 AND num(i,j,k)=res THEN 640
410 IF fila(i,j)=0 AND num(i,j,k)=res AND linea=0 THEN GOSUB 600
420 PRINT#i," ";v$;v$;v$;:num(i,j,k)=0:GOTO 440
430 PRINT#i," ";:PRINT#i,USING "###";num(i,j,k);
440 NEXT k:PRINT#i," ":PRINT#i," ":NEXT j,i:b$=INKEY$:b$=UPPER$(b$)
450 REM ----- TURNO DEL USUARIO (JUGADOR NUM. 4) -----
460 IF b$="L" OR b$="B" THEN 480
470 IF b$="" OR (b$<>"S") THEN 280
480 CLS#4:PRINT#4," "
490 FOR j=1 TO 3
500 FOR k=1 TO 5
510 IF num(4,j,k)<>res AND num(i,j,k)<>0 THEN 560
520 IF num(i,j,k)=res THEN fila(i,j)=fila(i,j)-1:tot(i)=tot(i)-1
530 IF tot(i)=0 AND num(i,j,k)=res AND b$="B" THEN 640
540 IF fila(i,j)=0 AND num(i,j,k)=res AND linea=0 AND b$="L" THEN GOSUB 600
550 PRINT#4," ";v$;v$;v$;:num(i,j,k)=0:GOTO 570
560 PRINT#4," ";:PRINT#i,USING "###";num(4,j,k);
570 NEXT k:PRINT#4," ":PRINT#4," ":NEXT j
580 GOTO 280
590 REM ----- SUBROUTINA INDICATIVA DE 'LINEA' -----
600 CLS#5:PRINT#5,"JUGADOR NUMERO: ";i;" HA CANTADO LINEA":SOUND 5,15,15
610 FOR z=1 TO 1000:NEXT:linea=1:IF i=4 THEN cant=cant*2
620 RETURN
630 REM ----- SUBROUTINA DE 'BINGO' -----
640 CLS#5:PRINT#5,"EL JUGADOR NUMERO: ";i;"HA CANTADO BINGO "
650 GOSUB 810
660 SOUND 10,10,10:FOR z=1 TO 2000:NEXT
670 IF i=4 THEN cant=cant*10
680 cant=cant-100:IF cant<=0 THEN RUN 40
690 REM ----- PANTALLA FINAL -----
700 CLS:PRINT "AUN QUEDAN ";cant;" PESETAS"
710 INPUT "Quieres seguir jugando ";y$:CLS
720 IF y$<>"S" AND y$<>"s" THEN 760
730 FOR i=1 TO 100:numero(i)=0:NEXT
740 linea=0
750 GOTO 130
760 END
770 REM ----- PANTALLA INICIAL -----
780 CLS:INPUT "Cantidad a jugar (100 Pts. carton)",c$
790 cant=VAL(c$):IF cant=0 OR cant MOD 100<>0 THEN 780
800 RETURN
810 READ nota,dur:IF nota=0 THEN RETURN
820 SOUND 1,nota/2,dur,7:SOUND 1,0,2
830 SOUND 4,nota*2,dur,6:SOUND 4,0,2
840 GOTO 810
850 DATA 322,30,287,30,256,30,242,60,322,90,242,30,256,30,242,30,215,60,287,90,
287,30,256,30,242,30,192,45,215,12,215,30,242,27,242,
30,256,30,287,30,256,30,242,120,0,0
```